
Change in Annex A

Annex A (normative):
IDL specifications

A.1
IDL specification (file name "CSIRPConstDefs.idl")
// File: CSIRPConstDefs.idl

#ifndef _CSIRPCONSTDEFS_IDL_

#define _CSIRPCONSTDEFS_IDL_

// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

/* ## Module: CSIRPConstDefs

This module contains commonly used definitions for CSIRP.

=========================================================

*/

module CSIRPConstDefs

{

  typedef unsigned short HeartbeatPeriodType;

  /*

  If notifyHeartbeat is triggered by NM positively by invoking 

  triggerHeartbeat operation, the value of this parameter shall be IRPManager,

  otherwise, it shall be IRPAgent.

  */

  enum TriggerFlagType {IRPManager, IRPAgent};

  typedef string ManagerIdentifierType;

  typedef string ChannelIdType;

  /*

  It specifies whether the operation is success or failed.

  */

  enum ResultType { Success, Failure };

   /**

   * This block identifies attributes which are included as part of the

   * CommunicationSurveillanceIRP. These attribute values should not 

   * clash with those defined for the attributes of notification 

   * header (see IDL of Notification IRP).

   */

   interface AttributeNameValue

   {

     const string HEARTBEAT_PERIOD = "HEARTBEAT_PERIOD";

     const string CHANNEL_ID = "CHANNEL_ID";

     const string TRIGGER_FLAG = "TRIGGER_FLAG";

     const string MANAGER_IDENTIFIER = "MANAGER_IDENTIFIER";

   };

};

#endif  // _CSIRPCONSTDEFS_IDL_
A.2
IDL specification (file name "CSIRPSystem.idl")

// File: CSIRPSystem.idl

#ifndef _CSIRPSYSYEM_IDL_

#define _CSIRPSYSYEM_IDL_

#include "ManagedGenericIRPSystem.idl"

#include "ManagedGenericIRPConstDefs.idl"

#include "CSIRPConstDefs.idl"

// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

/* ## Module: CSIRPSystem

This module implements capabilities of CSIRP.

=============================================

*/

module CSIRPSystem

{

  /** 

  * The InvalidHeartbeatPeriod exception is used when the period

  * value to be set by IRPManager is not a reasonable in IRPAgent's 

  * implementation. A very short period may cause IRPAgent to 

  * send many heartbeat notification in a short time, which may

  * decrease the performance of IRPAgent. To prevent this, 

  * IRPAgent may set the lower limit period in its system

  * implemntation. When the period to be set is shorter the

  * lower limit period, IRPAgent may throw this exception

  * and reject to set the period to new value.

  * Note: set the period to zero must be allowed. The behaviour of

  * setting period to zero pls see definition for Period.

  */

  exception InvalidHeartbeatPeriod

  {

    unsigned short periodLowerLimit;

    string reason;

  };

  exception InvalidManagerIdentifier { string reason; };
  exception ConflictingHeartbeatPeriod { string reason; };
  /*

   System fails to complete the operation.  System can provide reason

   to qualify the exception.  The semantics carried in reason

   is outside the scope of this IRP.

  */

  exception GetHeartbeatPeriod { string reason; };

  exception SetHeartbeatPeriod { string reason; };

  exception TriggerHeartbeat { string reason; };

  exception GetCSIRPVersions { string reason; };

  exception GetCSIRPOperationsProfile { string reason; };

  exception GetCSIRPNotificationProfile { string reason; };

   interface CSIRP

   {

     /**

     * IRPManager invokes this operation to obtain the current 

     * heartbeat period.

     */

     CSIRPConstDefs::ResultType get_heartbeat_period(

       out CSIRPConstDefs::HeartbeatPeriodType heartbeatPeriod

     )

     raises (GetHeartbeatPeriod);

     /**

     * IRPManager invokes this operation to set the heartbeatPeriod.

     * If the heartbeatPeriod is modified by one IRPManager, a

     * notifyHeartbeat notification should be emitted

     * immediately to all the subscribed IRPManagers to indicate

     * the new heartbeatPeriod. If the heartbeatPeriod is set to

     * zero, one notifyHeartbeat notification will be

     * emitted immediately and no more 

     * notifications unless the heartbeatPeriod is modified again.

     */

     CSIRPConstDefs::ResultType set_heartbeat_period(

       in CSIRPConstDefs::HeartbeatPeriodType heartbeatPeriod

     )

     raises (SetHeartbeatPeriod,

             ConflictingHeartbeatPeriod,

             InvalidHeartbeatPeriod,

             ManagedGenericIRPSystem::ValueNotSupported,

             ManagedGenericIRPSystem::OperationNotSupported);

     /*

     * IRPManager invoke this operation to trigger ET_HEARTBEAT 

     * notification positively.

     */

     CSIRPConstDefs::ResultType trigger_heartbeat(

       in CSIRPConstDefs::ManagerIdentifierType managerIdentifier           

     )

     raises (TriggerHeartbeat, InvalidManagerIdentifier);

     /**

     * Return the list of all supported CSIRP versions.

     */

     ManagedGenericIRPConstDefs::VersionNumberSet get_CS_IRP_versions (

     )

     raises (GetCSIRPVersions);

     /**

     * Return the list of all supported operations and their supported

     * parameters for a specific CSIRP version.

     */

     ManagedGenericIRPConstDefs::MethodList get_CS_IRP_operations_profile (

         in ManagedGenericIRPConstDefs::VersionNumber iRPVersion

     )

     raises (GetCSIRPOperationsProfile,

             ManagedGenericIRPSystem::OperationNotSupported,

             ManagedGenericIRPSystem::InvalidParameter);

     /**

     * Return the list of all supported notifications and their supported

     * parameters for a specific CSIRP version.

     */

     ManagedGenericIRPConstDefs::MethodList get_CS_IRP_notification_profile (

        in ManagedGenericIRPConstDefs::VersionNumber iRPVersion

     )

     raises (GetCSIRPNotificationProfile,

             ManagedGenericIRPSystem::OperationNotSupported,

             ManagedGenericIRPSystem::InvalidParameter);

   };     

};

#endif // _CSIRPSYSTEM_IDL_
A.3
IDL specification (file name "CSIRPNotifications.idl")
// File: CSIRPNotifications.idl

#ifndef _CSIRPNOTIFICATIONS_IDL_

#define _CSIRPNOTIFICATIONS_IDL_

#include "CSIRPConstDefs.idl"
#include "NotificationIRPConstDefs.idl"

#include "NotificationIRPNotifications.idl"
// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

/* ## Module: CSIRPNotifDefs
This module contains the specification of all notifications of CS IRP Agent.
============================================================================

*/

module CSIRPNotifications
{

   /**

   *  Constant definitions for the FileReady notification

   */      

   interface notifyHeartbeat: NotificationIRPNotifications::Notify

   {

     const string EVENT_TYPE = "notifyHeartbeat";
     /**

     * This constant defines the name of the  period property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property

     * is CSIRPConstDefs::HeartbeatPeriodType.

     */             

     const string HEARTBEAT_PERIOD = CSIRPConstDefs::AttributeNameValue::HEARTBEAT_PERIOD;

     /*

     * This constant defines the name of the 

     * channelId property, 

     * which is transported in the filterable_body

     * fields.

     * The data type for the value of this property

     * is CSIRPConstDefs::ChannelIdType.

     */             

     const string CHANNEL_ID = CSIRPConstDefs::AttributeNameValue::CHANNEL_ID;
     /*

     * This constant defines the name of the 

     * triggerFlag property, 

     * which is transported in the filterable_body

     * fields.

     * The data type for the value of this property

     * is CSIRPConstDefs::TriggerFlagType.

     */             

      const string TRIGGER_FLAG = CSIRPConstDefs::AttributeNameValue::TRIGGER_FLAG;

     /*

     * This constant defines the name of the 

     * managerIdentifier property, 

     * which is transported in the filterable_body

     * fields.

     * The data type for the value of this property

     * is CSIRPConstDefs::ManagerIdentifierType.

     */             

     const string MANAGER_IDENTIFIER = CSIRPConstDefs::AttributeNameValue::MANAGER_IDENTIFIER;
   };

};

#endif // _CSIRPNOTIFICATIONS_IDL_
End of change in Annex A

